Elementary maths

```
ans = Expand[(a+b)^3]                               (* mma (ans.101,ans) *)
ans = Factor[-2*x+2*x+a*x-x^2+a*x^2-x^3]            (* mma (ans.102,ans) *)
ans = Solve[x^2-4 == 0,x]                           (* mma (ans.103,ans) *)
ans = Solve[{2*a-b == 3, a+b+c == 1,-b+c == 6},{a,b,c}][[1]] (* mma (ans.104,ans) *)
ans = N[Pi,50]                                      (* mma (ans.105,ans) *)
ans = Apart[1/((1 + x) (5 + x))]                    (* mma (ans.106,ans) *)
ans = Together[(1/(1 + x) - 1/(5 + x))/4]           (* mma (ans.107,ans) *)
ans = Simplify[Tanh[Log[x]]]                        (* mma (ans.108,ans) *)
ans = Simplify[Tanh[I x]]                           (* mma (ans.109,ans) *)
ans = Simplify[Sinh[3 x] - 3 Sinh[x] - 4 (Sinh[x])^3] (* mma (ans.110,ans) *)
ans = HoldForm[Tanh[Log[x]]]                        (* mma (lhs.108,ans) *)
ans = HoldForm[Tanh[I x]]                           (* mma (lhs.109,ans) *)
ans = HoldForm[Sinh[3 x] - 3 Sinh[x] - 4 (Sinh[x])^3] (* mma (lhs.110,ans) *)
```

```
\begin{align*}
   &\mma*{ans.101}\\
   &\mma*{ans.102}\\
   &\mma*{ans.103}\\
   &\mma*{ans.104}\\
   &\mma*{ans.105}\\
   &\mma*{ans.106}\\
   &\mma*{ans.107}\\
   \mma{lhs.108} &= \Mma{ans.108}\\
   \mma{lhs.109} &= \Mma{ans.109}\\
   \mma{lhs.110} &= \Mma{ans.110}
\end{align*}
```

$$\texttt{ans.101} := a^3 + 3a^2b + 3ab^2 + b^3$$

$$\texttt{ans.102} := x(x+1)(a-x)$$

$$\texttt{ans.103} := \{\{x \to -2\}, \{x \to 2\}\}$$

$$\texttt{ans.104} := \left\{a \to \frac{1}{5}, b \to -\frac{13}{5}, c \to \frac{17}{5}\right\}$$

$$\texttt{ans.105} := 3.1415926535897932384626433832795028841971693993751$$

$$\texttt{ans.106} := \frac{1}{4(x+1)} - \frac{1}{4(x+5)}$$

$$\texttt{ans.107} := \frac{1}{(x+1)(x+5)}$$

$$\tanh(\log(x)) = \frac{x^2-1}{x^2+1} \tag{ans.108}$$

$$\tanh(ix) = i\tan(x) \tag{ans.109}$$

$$\sinh(3x) - 3\sinh(x) - 4\sinh^3(x) = 0 \tag{ans.110}$$

Linear Algebra

```
mat = {{2, 3}, {5, 4}}                       (* mma (ans.201,mat) *)
ans = Eigenvalues[mat]                       (* mma (ans.202,ans) *)
ans = Eigenvectors[mat]                      (* mma (ans.203,ans) *)
ans = CharacteristicPolynomial[mat, x]       (* mma (ans.204,ans) *)
rhs = {3, 7}
foo = rhs//MatrixForm                        (* mma (ans.205,foo) *)
sol = LinearSolve[mat,rhs]                   (* mma (ans.206,sol) *)
```

```
\begin{align*}
   &\mma*{ans.201}\\
   &\mma*{ans.202}\\
   &\mma*{ans.203}\\
   &\mma*{ans.204}\\
   &\mma*{ans.205}\\
   &\mma*{ans.206}
\end{align*}
```

$$\begin{aligned}
&\texttt{ans.201} := \begin{pmatrix} 2 & 3 \\ 5 & 4 \end{pmatrix}\\
&\texttt{ans.202} := \{7, -1\}\\
&\texttt{ans.203} := \begin{pmatrix} 3 & 5 \\ -1 & 1 \end{pmatrix}\\
&\texttt{ans.204} := x^2 - 6x - 7\\
&\texttt{ans.205} := \begin{pmatrix} 3 \\ 7 \end{pmatrix}\\
&\texttt{ans.206} := \left\{\frac{9}{7}, \frac{1}{7}\right\}
\end{aligned}$$

Limits

```
ans = Limit[Sin[4 x]/x,x->0]                (* mma (ans.301,ans) *)
ans = Limit[2^x/x,x->Infinity]              (* mma (ans.302,ans) *)
ans = Limit[((x+dx)^2 - x^2)/dx, dx -> 0]   (* mma (ans.303,ans) *)
ans = Limit[(4 n + 1)/(3 n - 1),n->Infinity] (* mma (ans.304,ans) *)
ans = Limit[(1+(a/n))^n,n->Infinity]        (* mma (ans.305,ans) *)
```

```
\begin{align*}
   &\mma*{ans.301}\\
   &\mma*{ans.302}\\
   &\mma*{ans.303}\\
   &\mma*{ans.304}\\
   &\mma*{ans.305}
\end{align*}
```

$$
\begin{aligned}
\texttt{ans.301} &:= 4\\
\texttt{ans.302} &:= \infty\\
\texttt{ans.303} &:= 2x\\
\texttt{ans.304} &:= \frac{4}{3}\\
\texttt{ans.305} &:= e^a
\end{aligned}
$$

Series

```
ans = Series[(1 + x)^(-2), {x, 1, 5}]        (* mma (ans.401,ans) *)
ans = Series[Exp[x], {x, 0, 5}]              (* mma (ans.402,ans) *)
ans = Sum[1/i^2, {i, 1, 50}]                 (* mma (ans.403,ans) *)
ans = Sum[1/i^4, {i, 1, Infinity}]           (* mma (ans.404,ans) *)
```

```
\begin{align*}
   &\mma*{ans.401}\\
   &\mma*{ans.402}\\
   &\mma*{ans.403}\\
   &\mma*{ans.404}
\end{align*}
```

$$
\begin{aligned}
\texttt{ans.401} &:= \frac{1}{4} - \frac{x-1}{4} + \frac{3}{16}(x-1)^2 - \frac{1}{8}(x-1)^3 + \frac{5}{64}(x-1)^4 - \frac{3}{64}(x-1)^5 + O\left((x-1)^6\right)\\
\texttt{ans.402} &:= 1 + x + \frac{x^2}{2} + \frac{x^3}{6} + \frac{x^4}{24} + \frac{x^5}{120} + O\left(x^6\right)\\
\texttt{ans.403} &:= \frac{3121579929551692678469635660835626209661709}{1920815367859463099600511526151929560192000}\\
\texttt{ans.404} &:= \frac{\pi^4}{90}
\end{aligned}
$$

Calculus

```
ans = D[x Sin[x],x]                                    (* mma (ans.501,ans) *)
ans = D[x Sin[x],x]/.x -> Pi/4                         (* mma (ans.502,ans) *)
ans = Integrate[2 Sin[x]^2, {x, a, b}]                 (* mma (ans.503,ans) *)
ans = Integrate[2 Exp[-x^2],{x, 0, Infinity}]          (* mma (ans.504,ans) *)
ans = HoldForm[Integrate[2 Exp[-x^2], {x, 0, Infinity}]]    (* mma (lhs.504,ans) *)
ans = Integrate[x^2 + y^2,  {x, 0, 1}, {y, 0, x}]      (* mma (ans.505,ans) *)
ans = HoldForm[Integrate[x^2 + y^2,  {x, 0, 1}, {y, 0, x}]] (* mma (lhs.505,ans) *)
```

```
\begin{align*}
   &\mma*{ans.501}\\
   &\mma*{ans.502}\\
   &\mma*{ans.503}\\
   \mma{lhs.504}&=\Mma{ans.504}\\
   \mma{lhs.505}&=\Mma{ans.505}
\end{align*}
```

$$\texttt{ans.501} := \sin(x) + x\cos(x)$$

$$\texttt{ans.502} := \frac{1}{\sqrt{2}} + \frac{\pi}{4\sqrt{2}}$$

$$\texttt{ans.503} := -a + \sin(a)\cos(a) + b - \sin(b)\cos(b)$$

$$\int_0^\infty 2\exp\left(-x^2\right)\, dx = \sqrt{\pi} \qquad (\texttt{ans.504})$$

$$\int_0^1 \int_0^x \left(x^2 + y^2\right) dy dx = \frac{1}{3} \qquad (\texttt{ans.505})$$

Differential equations

```
sol = DSolveValue[y'[x] + y[x] == 2 a Sin[x], y[x], x]                    (* mma (ans.601,sol) *)
sol = DSolveValue[{y'[x] + y[x] == 2 a Sin[x], y[0] == 0}, y[x], x]       (* mma (ans.602,sol) *)
sol = DSolveValue[{y''[x] + y[x] == 0}, y[x], x]                          (* mma (ans.603,sol) *)
sol = DSolveValue[{y''[x] + y[x] == 0, y[0] == 0, y'[0] == 1}, y[x], x]   (* mma (ans.604,sol) *)
ans = DSolveValue[y''[x] + 5 y'[x] - 6 y[x] == 0, y[x], x]                (* mma (ans.605,ans) *)
foo = ans /. {C[1] -> 2, C[2] -> 3}                                       (* mma (ans.606,foo) *)
```

```
\begin{align*}
   &\mma*{ans.601}\\
   &\mma*{ans.602}\\
   &\mma*{ans.603}\\
   &\mma*{ans.604}\\
   &\mma*{ans.605}\\
   &\mma*{ans.606}
\end{align*}
```

$$
\begin{aligned}
\texttt{ans.601} &:= a(\sin(x) - \cos(x)) + c_1 e^{-x}\\
\texttt{ans.602} &:= -a e^{-x}\left(-e^x \sin(x) + e^x \cos(x) - 1\right)\\
\texttt{ans.603} &:= c_2 \sin(x) + c_1 \cos(x)\\
\texttt{ans.604} &:= \sin(x)\\
\texttt{ans.605} &:= c_1 e^{-6x} + c_2 e^x\\
\texttt{ans.606} &:= 2e^{-6x} + 3e^x
\end{aligned}
$$

A table of derivatives and anti-derivatives

This example is based upon a nice example in the Pythontex gallery, see https://github.com/gpoore/pythontex/. It uses a tagged block to capture the Mathematica output for later use in the body of the LaTeX table.

```
(* Create a list of functions to include in the table *)
fun = {Sin[x],      Cos[x],      Tan[x],
       ArcSin[x],   ArcCos[x],   ArcTan[x],
       Sinh[x],     Cosh[x],     Tanh[x]};

eol = {"\\\\" ,     "\\\\",      "\\\\",
       "\\\\[5pt]", "\\\\[5pt]", "\\\\[5pt]",
       "\\\\",      "\\\\",      " "};

ddxfun = D[#, x] & /@ fun;
intfun = Integrate[#, x] & /@ fun;

ddxfunHold = HoldForm[D[#, x]] & /@ fun;
intfunHold = HoldForm[Integrate[#, x]] & /@ fun;

(* mmaBeg (CalculusTable) *)
Do[Print[OutputForm[
   ToString[TeXForm[ddxfunHold[[i]]]] <> "&=" <>
   ToString[TeXForm[ddxfun[[i]]]]     <> "\\quad & \\quad"
   ToString[TeXForm[intfunHold[[i]]]] <> "&=" <>
   ToString[TeXForm[intfun[[i]]]]     <>
   eol[[i]]
   ]], {i,1,9}]
(* mmaEnd (CalculusTable) *)
```

```
\begin{align*}
   \mma {CalculusTable}
\end{align*}
```

$$\frac{\partial \sin(x)}{\partial x} = \cos(x)$$
$$\frac{\partial \cos(x)}{\partial x} = -\sin(x)$$
$$\frac{\partial \tan(x)}{\partial x} = \sec^2(x)$$
$$\frac{\partial \sin^{-1}(x)}{\partial x} = \frac{1}{\sqrt{1-x^2}}$$
$$\frac{\partial \cos^{-1}(x)}{\partial x} = -\frac{1}{\sqrt{1-x^2}}$$
$$\frac{\partial \tan^{-1}(x)}{\partial x} = \frac{1}{x^2+1}$$
$$\frac{\partial \sinh(x)}{\partial x} = \cosh(x)$$
$$\frac{\partial \cosh(x)}{\partial x} = \sinh(x)$$
$$\frac{\partial \tanh(x)}{\partial x} = \operatorname{sech}^2(x)$$

$$\int \sin(x)\,dx = -\cos(x)$$
$$\int \cos(x)\,dx = \sin(x)$$
$$\int \tan(x)\,dx = -\log(\cos(x))$$
$$\int \sin^{-1}(x)\,dx = \sqrt{1-x^2} + x\sin^{-1}(x)$$
$$\int \cos^{-1}(x)\,dx = x\cos^{-1}(x) - \sqrt{1-x^2}$$
$$\int \tan^{-1}(x)\,dx = x\tan^{-1}(x) - \frac{1}{2}\log\left(x^2+1\right)$$
$$\int \sinh(x)\,dx = \cosh(x)$$
$$\int \cosh(x)\,dx = \sinh(x)$$
$$\int \tanh(x)\,dx = \log(\cosh(x))$$

Step-by-step integration

This is another nice example drawn from the Pythontex gallery, see https://github.com/gpoore/pythontex.

It shows the step-by-step computations of a simple triple integral.

```
xmax = 2;  ymax = 3;  zmax = 4;
xmin = 0;  ymin = 0;  zmin = 0;

fun = f[x,y,z];

mytmp = HoldForm[Integrate[#1, {z, #6, #7},
                               {y, #4, #5},
                               {x, #2, #3}]] & @@ {fun, xmin, xmax, ymin, ymax, zmin, zmax};   (* mma(lhs.01,mytmp) *)

fun = x y + y Sin[z] + Cos[x+y];

myint = HoldForm[Integrate[#1, {z, #6, #7},
                               {y, #4, #5},
                               {x, #2, #3}]] & @@ {fun, xmin, xmax, ymin, ymax, zmin, zmax};   (* mma(rhs.01,myint) *)

myansx =          Integrate[#1, {x, #2, #3}] & @@ {fun, xmin, xmax};
myintx = HoldForm[Integrate[#1, {z, #4, #5},
                                {y, #2, #3}]] & @@ {myansx, ymin, ymax, zmin, zmax};           (* mma(rhs.02,myintx) *)

myansxy =          Integrate[#1, {y, #2, #3}]  & @@ {myansx, ymin, ymax};
myintxy = HoldForm[Integrate[#1, {z, #2, #3}]] & @@ {myansxy, zmin, zmax};                     (* mma(rhs.03,myintxy) *)

myansxyz = Integrate[#1, {z, #2, #3}] & @@ {myansxy, zmin, zmax};                              (* mma(rhs.04,myansxyz) *)

myapprox = N[myansxyz,15];                                                                     (* mma(rhs.05,myapprox) *)
```

$$
\begin{aligned}
\int_0^4 \int_0^3 \int_0^2 f(x,y,z)dxdydz &= \int_0^4 \int_0^3 \int_0^2 (xy + \cos(x+y) + y\sin(z))dxdydz \\
&= \int_0^4 \int_0^3 2(y + \cos(1+y)\sin(1) + y\sin(z))dydz \\
&= \int_0^4 (8 + \cos(2) + \cos(3) - \cos(5) + 9\sin(z))\, dz \\
&= 41 + 4\cos(2) + 4\cos(3) - 9\cos(4) - 4\cos(5) \\[8pt]
&\approx 40.1235865133293
\end{aligned}
$$

```
\begin{align*}
   \mma{lhs.01} &= \mma{rhs.01}\\
                &= \mma{rhs.02}\\
                &= \mma{rhs.03}\\
                &= \mma{rhs.04}\\[8pt]
                &\approx \mma{rhs.05}
\end{align*}
```

Plotting Bessel functions

This simple example uses Mathematica to produce a plot of the first six Bessel functions. Two plots are shown, one created by Mathematica and a second created by LaTeX using the plotting package `pgfplots` and the data exported from Mathematica.

```
myData = Partition[Flatten[Table[{x, Table[BesselJ[n, x], {n, 0, 5}]}, {x, 0, 15, 0.1}]], 7];
myPlot = Plot[Evaluate[Table[BesselJ[n, x], {n, 0, 5}]], {x, 0, 15}, PlotLegends -> "Expressions"];

Export["example-04-fig.png", myPlot, "PNG"];
Export["example-04-fig.pdf", myPlot, "PDF"];
Export["example-04.txt", myData, "Table",  "FieldSeparators" -> " "];
```



Figure 1: The first six Bessel functions.

Using pgfplots



Figure 2: The first six Bessel functions.

```
\begin{tikzpicture} % requires \usepackage{pgfplots}
   \begin{axis}
      [xmin= 0.0,  xmax=15.0,
       ymin=-0.45, ymax=1.05,
       xlabel=$x$, ylabel=$J_n(x)$,
       grid=major, grid style={dashed,gray!30},
       legend entries = {$J_0$, $J_1$, $J_2$, $J_3$, $J_4$, $J_5$}]
       \addplot[blue]   table [x index=0, y index=1]{example-04.txt};
       \addplot[red]    table [x index=0, y index=2]{example-04.txt};
       \addplot[green]  table [x index=0, y index=3]{example-04.txt};
       \addplot[teal]   table [x index=0, y index=4]{example-04.txt};
       \addplot[orange] table [x index=0, y index=5]{example-04.txt};
       \addplot[purple] table [x index=0, y index=6]{example-04.txt};
   \end{axis}
\end{tikzpicture}
\captionof{figure}{The first six Bessel functions.} % requires \usepackage{caption}
```

Displaying long expressions

This example uses a simple (though contrived) example of a Taylor series expansion of $1/(1+x)$ to demonstrate the problems that can arise when displaying very long expressions.

```
f[x_] = 1/(1+x)
ans = f[x]                       (* mma (ans.511,ans) *)
ans = Series[f[x],{x, 0, 10}]    (* mma (ans.512,ans) *)
ans = Series[f[x],{x, 0, 20}]    (* mma (ans.513,ans) *)
ans = Series[f[x],{x, 0, 23}]    (* mma (ans.514,ans) *)
```

```
\begin{dgroup*}[spread={5pt}]
   \begin{dmath*} f(x) = \Mma*{ans.511} \end{dmath*}
   \begin{dmath*}    {}= \Mma*{ans.512} \end{dmath*}
   \begin{dmath*}    {}= \Mma*{ans.513} \end{dmath*}
   \begin{dmath*}    {}= \Mma*{ans.514} \end{dmath*}
   \begin{dmath*}    {}= \Mma*[\hskip 2cm]{ans.514} \end{dmath*}
\end{dgroup*}
```

The first four lines of the following output were set using `\Mma*` while the final line used `\Mma*[\hskip=2cm]`. The last pair of lines displays the output for the same tag `ans.514` and clearly the formatting of the second last line is not ideal as the text has overlapped the tag. This was corrected in the final line by using the option argument `[\hskip=2cm]` in the call to `\Mma*`.

$$f(x) = \frac{1}{x+1} \tag{ans.511}$$

$$= 1 - x + x^2 - x^3 + x^4 - x^5 + x^6 - x^7 + x^8 - x^9 + x^{10} + O\left(x^{11}\right) \tag{ans.512}$$

$$= 1 - x + x^2 - x^3 + x^4 - x^5 + x^6 - x^7 + x^8 - x^9 + x^{10} - x^{11} + x^{12} - x^{13} + x^{14} - x^{15} + x^{16} - x^{17} + x^{18} - x^{19} + x^{20} + O\left(x^{21}\right) \tag{ans.513}$$

$$= 1 - x + x^2 - x^3 + x^4 - x^5 + x^6 - x^7 + x^8 - x^9 + x^{10} - x^{11} + x^{12} - x^{13} + x^{14} - x^{15} + x^{16} - x^{17} + x^{18} - x^{19} + x^{20} - x^{21} + x^{22} - x^{23} + O\left(x^{24}\right) \tag{ans.514}$$

$$= 1 - x + x^2 - x^3 + x^4 - x^5 + x^6 - x^7 + x^8 - x^9 + x^{10} - x^{11} + x^{12} - x^{13} + x^{14} - x^{15} + x^{16} - x^{17} + x^{18} - x^{19} + x^{20} - x^{21} + x^{22} - x^{23} + O\left(x^{24}\right) \tag{ans.514}$$

Quadratic convergence of Newton-Raphson iterations

This is a simple example that uses Mathematica to demonstrate the quadratic converegnce of Newton-Raphson iterations to the exact root of a non-linear equation.

```
f[x_] = N[x - Exp[-x] , 200];                                 (* work to 200 decimal digits *)
a = NestList[(# - f[#]/f'[#]) &, SetPrecision[1/2,200], 6];   (* list of x values *)
b = f /@ a;                                                   (* list of f values *)
c = #^2 & /@ b;                                               (* list of f^2 values *)
(* mmaBeg(table) *)
Print[OutputForm[
      ToString[0] <> "&" <>
      ToString[NumberForm[a[[1]], 25]] <> "&" <>
      ToString[ScientificForm[b[[1]], 11, NumberFormat -> (SequenceForm[#1, "e", #3] &)]] <>
      "\\\\"
      ]]
Do[Print[OutputForm[
         ToString[i-1] <> "&" <>
         ToString[NumberForm[a[[i]], 25]] <> "&" <>
         ToString[ScientificForm[b[[i]], 11, NumberFormat -> (SequenceForm[#1, "e", #3] &)]] <> "&" <>
         ToString[NumberForm[b[[i]]/c[[i - 1]], 5]] <>
         "\\\\"
         ]], {i, 2, 7}]
(* mmaEnd(table) *)
```

Note the clear quadratic convergence in the iterations – the last column settles to approximately -0.11546 independent of the number of iterations. This behaviour would not be seen using normal floating point computations as they are normally limited to no more than 18 decimal digits. This computation used 200 decimal digits.

Newton-Raphson iterations $x_{n+1} = x_n - f_n/f'_n\ ,\quad f(x) = x - e^{-x}$			
n	x_n	$\epsilon_n = x_n - e^{-x_n}$	$\epsilon_n/\epsilon_{n-1}^2$
0	0.50000000000000000000000000	-1.0653065971e-1	
1	0.56631100319721815304164 92	-1.3045098060e-3	-0.11495
2	0.56714316503486221278651 21	-1.9648047172e-7	-0.11546
3	0.56714329040978102869957 66	-4.4574262753e-15	-0.11546
4	0.56714329040978387299996 87	-2.2941072910e-30	-0.11546
5	0.56714329040978387299996 87	-6.0767705445e-61	-0.11546
6	0.56714329040978387299996 87	-4.2637434326e-122	-0.11546

```
\def\eps{\epsilon}
\def\RuleA{\vrule depth0pt  width0pt height14pt}
\def\RuleB{\vrule depth8pt  width0pt height14pt}
\def\RuleC{\vrule depth10pt width0pt height16pt}

\setlength{\tabcolsep}{0.025\textwidth}%

\begin{center}
\begin{tabular}{cccc}%
   \noalign{\hrule height 1pt}
   \multicolumn{4}{c}{\RuleC\rmfamily\bfseries%
   Newton-Raphson iterations \quad%
   $x_{n+1} = x_n - f_n/f'_n\ ,\quad f(x) = x-e^{-x}$}\\
   \noalign{\hrule height 1pt}
   \RuleB$n$&$ x_n$&$\eps_{n} =  x_{n} - e^{-x_{n}}$&$\eps_{n}/\eps_{n-1}^2$\\
   \noalign{\hrule height 0.5pt}
   \mma{table}
   \noalign{\hrule height 1pt}
\end{tabular}
\end{center}
```

Using tagged blocks

The following Mathematica code block contains a matched `mmaBeg/mmaEnd` pair, with the tag name `info`, to capture the output from the formatted Mathematica `Print` statements.

```
(* mmaBeg(info) *)
Print[OutputForm["Date : &"    <> ToString[DateString[]]<>"\\\\"]]
Print[OutputForm["System : &"  <> ToString[$System]<>"\\\\"]]
Print[OutputForm["Version : &" <> ToString[$Version]]]
(* mmaEnd(info) *)
```

```
\bgroup\tt
\begin{tabular}{rl}
   \mma{info}
\end{tabular}
\egroup
```

Here is the output caught from the above block.

Date :	Mon 27 Aug 2018 10:34:38
System :	Mac OS X x86 (64-bit)
Version :	11.3.0 for Mac OS X x86 (64-bit) (March 7, 2018)

A mixed Mathematica-Python example

This example demonstrates a cooperative effort where Mathematica is used to do the analytic computations while Python is used to plot the data.

The example chosen here is to find and plot the solution to the boundary value problem defined by

$$\frac{d^2y}{dx^2} + 2\frac{dy}{dx} + 10y = 0 \qquad \text{with } y(0) = 3,\ y'(0) = 0$$

This example requires two passes, once for Mathematica and once for Python (and in that order). This example can be run using

```
mmalatex.sh -x -i mixed
pylatex.sh  -x -i mixed
pdflatex          mixed
```

Note that the last pair of commands could also be combined as `pylatex.sh -i mixed`.

The Mathematica code

Here Mathematica is used to first find the general solution of th differential equation. The boundary conitions are then imposed and finally a uniform sampling of the solution is written to a file for later use by Python and Matplotlib.

```
sol = DSolve[y''[x] + 2 y'[x] + 10 y[x] == 0, y, x]
ans = y[x]/.sol[[1]]                      (* mma(ans.101,ans)*)

sol = DSolve[{y''[x] + 2 y'[x] + 10 y[x] == 0, y[0]==3, y'[0]==0}, y, x]
foo = y[x]/.sol[[1]]                      (* mma(ans.102,foo)*)
bah = Simplify[y'[x]/.sol[[1]]]           (* mma(ans.103,bah)*)

(* now sample y and dy at selected points *)
myData = Table[{x, foo, bah}, {x, 0, 2 Pi, 0.02}];
Export["mixed.txt", myData, "Table",  "FieldSeparators" -> " "];
```

The general solution of the differential equation is

$$y(x) = c_1 e^{-x}\sin(3x) + c_2 e^{-x}\cos(3x)$$

while the particular solution satifying the boundary conditions is given by

$$y(x) = e^{-x}(\sin(3x) + 3\cos(3x))$$

The Python code

This is a straighforward use of Matplotlib to plot two functions. The code reads the datafile created previously by Mathematica and then calls Matplotlib to plot that data.

```
import numpy as np
import matplotlib.pyplot as plt

plt.matplotlib.rc('text', usetex = True)
plt.matplotlib.rc('grid', linestyle = 'dotted')
plt.matplotlib.rc('figure', figsize = (5.5,4.1)) # (width,height) inches

x, y, dy = np.loadtxt ('mixed.txt', unpack=True)

plt.plot (x,y)
plt.plot (x,dy)

plt.xlim (0.0,4.0)

plt.legend(('$y(x)$', '$dy(x)/dx$'), loc = 0)
plt.xlabel('$x$')
plt.ylabel('$y(x),\> dy/dx$')
plt.grid(True)
plt.tight_layout(0.5)

plt.savefig('mixed-fig.pdf')
```



Figure 1: The function and its derivative.